Exploring Abstract Algebra with Computer Software
PREP Workshop 2004
Section 6: Factor Groups

GAP can be used to compute right cosets and factor groups.

```
gap> S6:= SymmetricGroup(6);
Sym( [1..6] )
gap> A6:= AlternatingGroup(6);
Alt( [1..6] )
D6:= DihedralGroup(IsPermGroup, 12);
Group([ (1,2,3,4,5,6), (2,6)(3,5) ])
gap> Z6 := Center(D6);
Group([ (1,4)(2,5)(3,6) ])
```

The above assigns the name `S6` to the symmetric group S_6, `A6` to the subgroup of `S6` of even permutations, `D6` to the dihedral group D_6, and `Z6` to the center of D_6. We use the option `IsPermGroup` for `D6` so that GAP constructs `D6` as a subgroup of `S6`. To form factor groups we need **normal** subgroups. Test which subgroups are normal:

```
gap> IsNormal(S6,A6);
true
gap> IsNormal(S6,D6);
false
gap> IsNormal(D6,Z6);
true
```

Thus A_6 is normal in S_6 and $Z(D_6)$ is normal in D_6.

```
gap> RightCosets(S6,A6);
[RightCoset(AlternatingGroup[ 1..6 ]), ()),
RightCoset(AlternatingGroup[ 1..6 ]), (5,6))]
```

The above output tells us the right cosets of A_6 in S_6 are A_6 and $A_6(5,6)$. Thus the factor group S_6/A_6 has two elements.

```
gap> Size(FactorGroup(S6,A6));
```

Now consider the factor group $D_6/Z(D_6)$.

```
gap> RightCosets(D6,Z6);
[ RightCoset(Group([(1,4)(2,5)(3,6)] ), ()),
RightCoset(Group([(1,4)(2,5)(3,6)] ), (2,6)(3,5)),
RightCoset(Group([(1,4)(2,5)(3,6)] ), (1,3)(4,6)),
RightCoset(Group([(1,4)(2,5)(3,6)] ), (1,3,5)(2,4,6)),
RightCoset(Group([(1,4)(2,5)(3,6)] ), (1,5)(2,4)),
RightCoset(Group([(1,4)(2,5)(3,6)] ), (1,5,3)(2,6,4)) ]
```

Thus there are 6 right cosets: N, $N(2,6)(3,5)$, $N(1,3)(4,6)$, $N(1,3,5)(2,4,6)$, $N(1,5)(2,4)$ and $N(1,5,3)(2,6,4)$ where $N = Z(D_6)$. So the factor group has 6 elements. Which group of order 6 is $D_6/Z(D_6)$? We will now use GAP to help us answer this question.

```
gap> F:= FactorGroup(D6,Z6);
Group([ f1, f2 ])
gap> IsAbelian(F);
false
```

Since $D_6/Z(D_6)$ is non-Abelian, it can not be isomorphic to $\mathbf{Z}_6$. Thus check to see if $D_6/Z(D_6)$ is isomorphic to S_3:

```
gap> IsomorphismGroups(F,SymmetricGroup(3));
[ f1, f2 ] -> [ (2,3), (1,2,3) ]
```

So $D_6/Z(D_6)$ is isomorphic to S_3.

GAP will also tell you which elements are in a particular coset. For example:

```
gap> Elements(RightCoset(Z6, (2,6)(3,5)));
[ (2,6)(3,5), (1,4)(2,3)(5,6)]
```

Thus the right coset $Z(D_6)(2,6)(3,5) = \{(2,6)(3,5), (1,4)(2,3)(5,6)\}$.

Project, Section 6

6.1 Use GAP to find the right cosets of $Z(D_8)$ in D_8.

6.2 By hand, write out the Cayley table of the factor group $D_8/Z(D_8)$.

You can check your work to Exercise 6.2 by using the GAP command `MultiplicationTable`. For example, to find the Cayley table for S_3 type:

```
gap> e:= Elements(SymmetricGroup(3));
[ (), (2,3), (1,2), (1,2,3), (1,3,2), (1,3) ]
gap> PrintArray(MultiplicationTable( e ));
[ [  1,  2,  3,  4,  5,  6 ],
  [  2,  1,  4,  3,  6,  5 ],
  [  3,  5,  1,  6,  2,  4 ],
  [  4,  6,  2,  5,  1,  3 ],
  [  5,  3,  6,  1,  4,  2 ],
  [  6,  4,  5,  2,  3,  1 ] ]
```

The GAP output of `PrintArray(MultiplicationTable( e ));` is an n by n array (where n is the order of the group) such that the integer in row i column j equal k if and only if the ith element in the list times the jth element equals the kth element.

6.3 The factor group $D_8/Z(D_8)$ is isomorphic to a group we have used often. Use GAP to help you determine which one.

6.4 Repeat Exercise 6.3 for the factor groups $D_{10}/Z(D_{10})$ and $D_{12}/Z(D_{12})$.

6.5 Based on your results in Exercises 6.3 and 6.4, make a conjecture about the factor group $D_n/Z(D_n)$ when n is even and greater than or equal to 8.